
Electronic Circuits
Lecture 6.1: Pspice Examples



Pspice Tutorial: Comments & End of File 

 * TITLE

 * comment as a whole line

 ; comment after a legal command to the end of the line

 The first line must be a title to describe the circuit!

 .END ; End of the simulation file



Pspice Tutorial: Sources 

 SOURCE, GENERAL
 Vxxx N1 N2 DC [VALUE]

 Vxxx N1 N2 AC [VALUE]

 Vxxx N1 N2 SIN(VO VA [Freq[Td[Df[Phase]]]])

 Vxxx N1 N2 EXP (V1 V2 [Td1[Td2[Tau2]]])

 Vxxx N1 N2 PWL (T1 V1 {Tu Vu]...)

 Vxxx N1 N2 PULSE (V1 V2 [Td[Tr[Tf[PW[Period]]]]])

 Ixxx N1 N2 DC [VALUE]

 Ixxx N1 N2 AC [VALUE]

 Ixxx N1 N2 SIN(VO VA [Freq[Td[Df[Phase]]]])

 Ixxx N1 N2 EXP (V1 V2 [Td1[Td2[Tau2]]])

 Ixxx N1 N2 PWL (T1 V1 {Tu Vu]...)

 Ixxx N1 N2 PULSE (V1 V2 [Td[Tr[Tf[PW[Period]]]]])

 INDEPENDENT SOURCES
 Vxxxxxxx N+ N- DC (PULSE,SIN,EXP) AC ; (voltage 

source)

 Ixxxxxxx N+ N- DC (PULSE,SIN,EXP) AC ; (current 
source)



Pspice Tutorial: Dependent Sources 

 CONTROLLED SOURCES (VCVS, ICIC, VCIS, ICVS)
 Exxxxxxx N+ N- NC+ NC- VALUE ; (voltage controlled voltage source)

 Fxxxxxxx N+ N- VNAME VALUE ; (current controlled current source)

 Gxxxxxxx N+ N- NC+ NC- VALUE ; (voltage controlled current source)

 Hxxxxxxx N+ N- VNAME VALUE ; (current controlled voltage source)



Pspice Tutorial: R-L-C Elements 

 RLC ELEMENTS
 Rxxxxxxx N1 N2 VALUE ; (Resistor)

 Cxxxxxxx N+ N- VALUE IC ; (Capacitor)

 Lxxxxxxx N+ N- VALUE IC ; (Inductor)

 Kzzzzzzz Lxxxxxxx Lyyyyyyy VALUE ; (Transformer)



Pspice Tutorial: Semiconductor Devices

 SEMICONDUCTOR DEVICES
 Dxxxxxxx N+ N- DMOD ; (diode)

 .MODEL DMOD D DIODE PARAMETER ; (diode model)

 Qxxxxxxx NC NB NE QMOD ; (bipolar transistor)

 .MODEL QMOD NPN or PNP TRANSISTOR PARAMETERS ; (transistor model)



Pspice Tutorial: Analysis Methods 

 ANALYSIS
 .DC SWEEP VARIABLE START STOP INCREMENT (DC)

 .TF OUTPUT VARIABLE INPUT SOURCE ; (transfer function)

 .TRAN TSTEP TSTOP UIC ; (transient)

 .AC DEC or OCT or LIN NPOINTS FSTART FSTOP (AC)

 .IC X(NODE) = VALUE

 .LIB NAME.LIB ; (library statement)



Pspice Tutorial: Reporting Outputs

 OUTPUT
 .PRINT (PLOT) DC OUTVAR1...OUTVAR8

 .PRINT (PLOT) TRAN OUTVAR1...OUTVAR8

 .PRINT (PLOT) AC VM VDB VP VR VI

 .PROBE



Pspice Tutorial: Prefixes and Suffixes

 F fento- 10-15

 P pico- 10-12

 N nano- 10-9

 U micro- 10-6

 M milli- 10-3

 K kilo- 10+3

 MEG mega- 10+6

 G giga- 10+9

 T tera- 10+12

 MIL (0.001") 25.4×10-6



Pspice Tutorial: Magnitude and Phase

 VM(n) - magnitude of nth node voltage

 VP(n) - phase of nth node voltage in degrees



Example 1: Find the current across each branch and voltage across each 
element.

 After associating ALL nodes (including non-essential 
ones) with numbers, with node 0 as the reference or 
ground node, the circuit elements must be 
labeled/associated as well by PSpice standard symbol 
representations. After doing this, the netlist which 
PSpice can understand can be written and simulated.

 So as you can see, one MUST know how to 
understand how circuit elements are arranged and 
assembled (according to polarity and direction) in a 
PSpice Netlist.



Example 1: PSpice solution

*BME206 - Example 1
R1 1 4 2ohm
R2 1 2 3ohm
R3 3 4 4ohm
R4 3 4 4.5ohm
R5 4 0 9ohm
V1 1 0 2.5V
V2 3 2 3V
I1 2 0 10A
.DC V1 10 10 1 ; analysis method
*To find ALL current and voltages across
.PRINT DC V(R1), V(R2), V(R3), V(R4), 
V(R5), V(V1), V(V2), V(I1), I(R1), I(R2), 
I(R3), I(R4), I(R5), I(V1), I(V2), I(I1)
.OPTIONS NOPAGE ; to minimize the output
.END ;must have to end netlist.



Example 2: Find the current across each branch and voltage across each 
element.

 The only difference between the examples is that two 
of the sources are now dependent sources. And since 
H1 is a current dependent voltage source, a voltmeter 
is required along the controlling branch which will let 
us extract the current through the resistor and use it 
through PSpice.



Example 2: PSpice solution

*BME206 - Example 2
R1 1 4 2ohm
R2 1 2 3ohm
R3 3 4 4ohm
R4 3 4 4.5ohm
R5 5 0 9ohm
Vm 4 5 ;Voltmeter used for H1
V1 3 2 3V
H1 1 0 Vm 2.5
G1 2 0 1 4 5
.DC V1 10 10 1 ; analysis method
.PRINT DC V(R1), V(R2), V(R3), V(R4), 
V(R5), V(V1), V(H1), V(G1), I(R1), I(R2), 
I(R3), I(R4), I(R5), I(V1), I(G1), I(H1)
.OPTIONS NOPAGE
.END



Example 3: Find the voltage VC(t) in the RC circuit below.

 Note that the "use initial condition" command is used 
in the TRAN statement and that the initial condition 
for the capacitor is specified. In this case, the applied 
voltage was assumed to be a step function, i.e. 
V1=10V t>0; thus a simple DC source was used to 
represent the input voltage. In other cases, a time 
dependent source such as a sinewave, pulse, etc., may 
be required.



Example 3: PSpice solution

*BME206 - Example 3
R1 1 2 50ohm
V1 1 0 10V
C1 2 0 0.02UF IC=0V ; cap w I. condition
.TRAN 0.05US 5US UIC ; analysis method
.PROBE ; PROBE graphics output, V(t)
.OPTIONS NOPAGE
.END



Example 4: Find the voltage VC(t) in the RC circuit below.

 Find the current iC(f), VR(f), and VC(f).

 AC analysis using PSPICE is phasor analysis: the 
source is an AC/Phasor source of amplitude 10V and 
phase of zero degrees.



Example 4: PSpice solution

R1 1 2 50ohm
V1 1 0 AC 10V 0 ; 10 volts, 0 degrees phase
C1 2 0 0.02UF ; capacitor w initial condition
.AC DEC 20 10KHZ 1MEGAHZ ; method
.PROBE ; PROBE graphics output, trace V(t)
.OPTIONS NOPAGE
.END



Example 5: Common Emitter Amplifier
* Common emitter amplifier

.model 2N3904  NPN(Is=6.734f Xti=3 Eg=1.11 Vaf=74.03 Bf=416.4 Ne=1.259

+              Ise=6.734f Ikf=66.78m Xtb=1.5 Br=.7371 Nc=2 Isc=0 Ikr=0 Rc=1

+              Cjc=3.638p Mjc=.3085 Vjc=.75 Fc=.5 Cje=4.493p Mje=.2593 

+              Vje=.75 Tr=239.5n Tf=301.2p Itf=.4 Vtf=4 Xtf=2 Rb=10)

* 

V1 Vcc 0 DC 12V 

* need extra 0 at "4th" connection of transistors to indicate the reference point

Q1 Vc Vb Ve 0 2N3904      

Rc Vcc Vc 2k

Re Ve 0 500

R1 Vb 0 7k

R2 Vcc Vb 33k

C1 Vin Vb 0.1u    

C2 Vc Vout 0.1u    

Rload Vout 0 100k    

V2 Vin 0 ac 10mV sin(0 10mV 10kHz) 

.tran 1u 5m

.ac oct 5 100 100Meg

.end

* .model is the part definition, 
you may not understand all 
parameters given in the model, 
no problem 
* You can name Re, instead of 
using numbers only as R2, etc.
* Here, Vin is the node name !



Thanks for 
listening 

YALÇIN İŞLER

Assoc. Prof.

